*=============Windowing_AS_AM===========

Hann window



$$w(n) = 0.5\left(1 - \cos\left(\frac{2\pi n}{N-1}\right)\right)$$

The Hann window function really is nothing more than 100% amplitude modulating the input signal by a cosine wave with a period exactly equal to the sample period.

To see what it does, first take a perfect 14Khz input signal and convert it to a spectrum without any windowing.

```
=======================================================================

VsigR      VsigR    0       DC      0       SIN(    0       1       14k       )

*TRAN      TSTEP TSTOP TSTART TMAX ?UIC?
tran       .1u    1m     0       .1u
set        specwindow=    "rectangular"
spec       1k          100k    1k      v(VsigR)
plot       dB(mag(VsigR))    xlog ylimit -120 20
```



Since the input signal contains whole number of cycles in the 1msec sample period, the spectrum will have a clean signal at frequency bin number 14.

Now redo the same thing, except use a hanning window.

==

```
tran        .1u    1m     0        .1u
set         specwindow=    "hanning"
spec        1k         100k    1k       v(VsigR)
plot        dB(mag(VsigR))    xlog ylimit -120 20
```



Lets sanity check the spectrum output by externally amplitude modulating the same signal with the Hann function, and redo the spectrum analysis again with a rectangular window.

==

```
VCos1       VCos1    0       DC      0       SIN(    0       1       1k      -.25m )
VsigR       VsigR    0       DC      0       SIN(    0       1       14k           )
BhannR      VhannR   0       V =     V(VsigR)*(.5-.5*v(VCos1))
```



The input signal will look like so. The output of the spectrum comes out very close to the perfect 14KHz signal being processed with a hanning window.

==

```
tran        .1u      1m       0        .1u
plot        VhannR
set         specwindow=    "rectangular"
spec        1k         100k    1k       v(VhannR)
plot        dB(mag(VhannR))   xlog    ylimit -120 20
```



The spectrum is actually a classical AM spectrum. There is a 14KHz carrier with -6dB side bands at 13KHz and 15KHz.

==



Since in this case the windowing is being done externally, it is possible to double the sample period, and there by double the frequency resolution.

==

```
tran      .1u       2m    0      .1u
plot      VhannR
set       specwindow=   "rectangular"
spec      .5k       100k   .5k      v(VhannR)
plot      dB(mag(VhannR))   xlog ylimit -120 20
plot      dB(mag(VhannR))   xlog ylimit -20 0
```





Now it is easy to see that there are clean signals at 13KHz, 14KHz, and 15KHz.

===



Now the 14kHz signal has been transformed to three frequencies. When thinking in terms of power, the two -6dB sidebands should increase the 14KHz power by about 22% or +1.76dB.

The Hanning window is effectly doing what is shown below.

==

http://www.idea2ic.com/PlayWithJavascript/3D_FFT8v.html



wt/wt_0	Real	Imaginary
0	0.000	0.000
1	0.000	0.000
2	0.000	0.000
3	0.000	0.000
4	0.000	0.000
5	0.000	0.000
6	0.000	0.000
7	0.000	0.000
8	0.000	0.000
9	0.000	0.000
10	0.000	0.000
11	0.000	0.000
12	0.000	0.000
13	0.2500	0.000
14	0.5000	0.000
15	0.2500	0.000
16	0.000	0.000

The Hanning windowing is simple amplitude modulation.

==





When looking at a hanning spectrum, every single frequency is going to be transformed into at least three frequencies. This should be taken into consideration when trying to extract a power value of such a spectrum. The side bands contribute about 22% to the total power.

===

```
tran    .1u    2m    0    .1u
set     specwindow =      "hanning"
spec    .5k        100k   .5k    v(VsigR)
plot    dB(mag(VsigR))    xlog ylimit -120 20
```



The wider spectrum can be narrowed a factor of two by just taking twice as long of a sample period. When this example used a 1msec sample period, the spectrum resolution is 1KHz, and the hanning windowed spectrum is spread out over three 1KHz frequency bins. When the sample period is 2msec, then the spectrum is spread out over 1.5KHz worth of frequency bins.

```
===========Full_Netlist_For_Copy_Paste=======================
Windows_As_AM
.Option srcsteps = 1 set Gmin = 1.0000E-02
*=========Circuit_Netlist=========================
*V_SIN#    NODE_P  NODE_N DC     VALUE  SIN(   V_DC    AC_MAG FREQ   DELAY  FDamp)
Vsig       Vsig    0      DC     0      SIN(   0       1      14.1k       )
VCos1      VCos1   0      DC     0      SIN(   0       1      1k     -.25m )
Bhann      Vhann   0      V =    V(Vsig)*(.5-.5*v(VCos1))

VsigR      VsigR   0      DC     0      SIN(   0       1      14k        )
BhannR     VhannR  0      V =    V(VsigR)*(.5-.5*v(VCos1))

.control
set        pensize = 2
*TRAN      TSTEP TSTOP TSTART TMAX ?UIC?
tran       .1u   1m    0      .1u
set        specwindow=   "rectangular"
spec       1k        100k   1k      v(VsigR)
plot       dB(mag(VsigR))   xlog ylimit -120 20

tran       .1u   1m    0      .1u
set        specwindow=   "hanning"
spec       1k        100k   1k      v(VsigR)
plot       dB(mag(VsigR))   xlog ylimit -120 20

tran       .1u   1m    0      .1u
plot       VhannR
set        specwindow=   "rectangular"
spec       1k        100k   1k      v(VhannR)
plot       dB(mag(VhannR))   xlog ylimit -120 20
plot       dB(mag(VhannR))   xlog ylimit -20 0

tran       .1u   2m    0      .1u
plot       VhannR
set        specwindow=   "rectangular"
spec       .5k       100k   .5k     v(VhannR)
plot       dB(mag(VhannR))   xlog ylimit -120 20
plot       dB(mag(VhannR))   xlog ylimit -20 0

.endc
.end
```

7.29.10_12.02PM

dsauersanjose@aol.com
Don Sauer
http://www.idea2ic.com/